An Introduction to *Maple*

Graphing Implicit Functions, Parametric Equations, and Polar Equations

by Bob Bradshaw, Ohlone College, Fremont, CA

Start by loading the plotting subroutines

```
> restart;with(plots);
>
```

Implicit Functions

An implicit function is one in which you do not solve the equation for y. The implicitplot command will allow you to graph this type of equation.

```
> implicitplot(x^2+y^2=1,x=-2..2,y=-2..2,thickness=3);
```

To make the ellipse look like a circle, I want to set the scaling option so that scaling = constrained. Since I want this to be true for all the graphs that follow, I set the option in a separate statement now, rather than retypring the command with every plot.

```
> setoptions(scaling=constrained);
  implicitplot(x^2+y^2=1,x=-2..2,y=-2..2,thickness=3);
```

Unfortunely, implicitplot can give an ugly graph.

```
> implicitplot(x^3*y+x^2*y^3=4,x=-10..10,y=-10..10);
```

Increasing the number of points helps but the graph is incorrect since this equation is a function!

```
> implicitplot(x^3*y+x^2*y^3=4,x=-10..10,y=-10..10,numpoints=1000
  );
```

Parametric Equations

Parametric equations describe the horizontal and vertical positions of a point on a curve by writing both x and y in terms of a third variable t, which often represents time. This material is usually covered in Calculus II. To graph parametric equations, define the x and y values and then use the plot command with the format plot([*x,y,t=start..end*]). You can use any of the options that are used with the graphs of standard functions.

```
> x:=cos(t);
  y:=sin(t);
  plot([x,y,t=0..2*Pi]);
```

```
> horiz:=t-2*sin(t);
> vert:=1-2*cos(t);
> plot([horiz,vert,t=0..8*Pi]);
```

Polar Coordinates

A polar coordinate system plots coordinates based on the distance of a point from the origin (the radius) and the angle formed by the x -axis and the line between the origin and the point.

There are three different commands for plotting in polar coordinates. I suggest that you use polarplot([*radius, angle, angle=start..end*], scaling=constrained).

To graph the curve $r = \cos(2\theta)$, use

```
> polarplot([cos(2*theta),theta,theta=0..2*Pi]);
```

A second method is to use the standard plot command but to forcce *Maple* to use polar coordinates.

```
> plot(cos(2*t),t=0..4*Pi,coords=polar);
```

The disadvantage of usinghe polarplot or standard plot commands is thatmost standard polar functions are centered around the origin. It requires some messy algebra to move the above graph so that it is centered around another point, say (2, 3).

To avoid this difficulty, we can use parametric equations. The relationship between a polar equation r and parametric equations is that $x = r\cos(\theta)$ and $y = r\sin(\theta)$. Consider the following. Noptice that the graph is now centered around (2, 3).

```
> r1:=cos(2*theta);
  x1:=2+r1*cos(theta);
  y1:=3+r1*sin(theta);
  plot([x1,y1,theta=0..2*Pi]);
```

We can now use this method to combine multpile polar plots.

```
> r1:=2*cos(2*theta);
  x1:=-2+r1*cos(theta);
  y1:=3+r1*sin(theta);
  firstcurve:=plot([x1,y1,theta=0..2*Pi],color=orange,thickness=3
  ):

  r2:=2*cos(3*theta);
  x2:=2+r2*cos(theta);
  y2:=3+r2*sin(theta);
  secondcurve:=plot([x2,y2,theta=0..2*Pi],color=green,thickness=4
  ):

  r3:=2*cos(6*theta);
  x3:=-2+r3*cos(theta);
  y3:=-3+r3*sin(theta);
  thirdcurve:=plot([x3,y3,theta=0..2*Pi],color=blue,thickness=1):

  r4:=1.5*cos(5*theta);
  x4:=2+r4*cos(theta);
  y4:=-3+r4*sin(theta);
  fourthcurve:=plot([x4,y4,theta=0..2*Pi],color=magenta,thickness
  =2):

  display([firstcurve,secondcurve,thirdcurve,fourthcurve],scaling
  =constrained,axes=boxed);
```

Our look at polar graphs in *Maple* would be incomplete without looking at the following:

```
> S := 100/(100+(t-Pi/2)^8):
  R := S*(2-sin(7*t)-cos(30*t)/2):
  plot([R,t,t=-Pi/2..3/2*Pi],coords=polar,numpoints=2000,axes=NON
  E);
>
```

You Try It!

(1) Graph the set of parametric equations
$x = 5\cos(t) + 14\cos(15t)$, $y = 5\sin(t) + 14\sin(15t)$

(2) Graph the polar function $r = t$ where $0 < t < 6*\mathrm{Pi}$

(3) Graph the implicit function $|x|^{\left(\frac{2}{3}\right)} + |y|^{\left(\frac{2}{3}\right)} = 1$ on the interval $-1 < x < 1$, $-1 < y < 1$. Use 1000 points and abs( ) for the absolute value function.

(4) Create a single graph containing a plot of an implicit, a parametric, and a polar equation.